

Работа с базой данных

Цель работы: Изучение подключения к базе данных, создание Query Table на основе объекта **RecordSet**, в котором хранится информация, полученная из базы данных.

Задание 1 Загрузка данных

В компании ведется учет товаров, которые имеются на складе, при помощи таблицы **Products** базы данных **Nwind**.

В этой таблице находятся следующие важные столбцы:

- *Product ID*—идентификатор товара;
- *ProductName*—наименование продукта;
- *UnitPrice* — стоимость продукта за единицу;
- *UnitsInStock (На складе)*—количество единиц этого товара на складе;
- *ReorderLevel (Минимальный запас)*— минимально допустимое количество единиц данного товара на складе. Если реальное количество единиц этого товара меньше, чем этот уровень, товар нужно срочно заказывать;
- *Discontinued (Поставки прекращены)*— флаг прекращения работы с товаром. Если в этом столбце стоит «истина», то это значит, что принято решение закупки этого товара больше не производить.

Все остальные столбцы для целей этой работы можно игнорировать.

Заполнение таблицы **Products** производится при помощи специализированного приложения, созданного достаточно давно и не предусматривающего некоторых необходимых форм.

Требуется

Создать приложение на основе Excel, которое:

- Производит вставку в лист Excel данные по всем строкам и всем вышеуказанным столбцам этой таблицы.
- Генерирует в Excel дополнительные столбцы следующего содержания:

Заказать товар, штук — разница между столбцами **ReorderLevel (Минимальный запас)** и **UnitsInStock (На складе)**. В этот столбец помещается информация о количестве товара в штуках, которое нужно срочно заказать.

Эту информацию нужно генерировать только для тех записей, для которых значение в столбце **ReorderLevel (Минимальный запас)** больше, чем в

столбце **UnitsInStock (На складе)**, и у которых значение столбца **Discontinued (Поставки прекращены)** установлено в Ложь.

Стоимость заказа— определяло бы стоимость такого пополнения склада для каждой строки в таблице. Стоимость заказа рассчитывается как произведение предыдущего столбца и столбца **UnitPrice (Цена)**.

Эту информацию также нужно генерировать только для тех записей, для которых значение в столбце **ReorderLevel (Минимальный запас)** больше, чем в столбце **UnitsInStock (На складе)**.

- Вставляло бы одной строкой ниже полученных записей из базы данных две итоговые строки: **общая стоимость товаров на складе** — итоговая стоимость всех товаров, которые находятся на складе (как сумма произведений столбцов **На складе** и **Цена** для каждой строки);

общая стоимость товаров к заказу— итог по столбцу **Стоимость заказа**.

Общий вид получившегося приложения может быть таким, как представлено на рис. 1:

ProductID	ProductName	UnitPrice	UnitsInStock	ReorderLevel	Discontinued	Заказать товара, штук	Стоимость заказа
1	Chai	18	39	10	ЛОЖЬ		
2	Chang	19	17	25	ЛОЖЬ	8	152,00р.
3	Aniseed Syrup	10	13	25	ЛОЖЬ	12	120,00р.
4	Chef Anton's Cajun Seasoning	22	53	0	ЛОЖЬ		
5	Chef Anton's Gumbo Mix	21,35	0	0	ИСТИНА		

Рис. 1 Первые строки листа с импортированными данными

Итоговые строки могут выглядеть так, как показано на рис. 2:

ProductID	ProductName	UnitPrice	UnitsInStock	ReorderLevel	Discontinued	Заказать товара, штук	Стоимость заказа
69	Gudbrandsdalsost	36	26	15	ЛОЖЬ		
70	Outback Lager	15	15	30	ЛОЖЬ	15	225,00р.
71	Flutemysost	21,5	26	0	ЛОЖЬ		
72	Mozzarella di Giovanni	34,8	14	0	ЛОЖЬ		
73	Rød Kaviar	15	101	5	ЛОЖЬ		
74	Longlife Tofu	10	4	5	ЛОЖЬ	1	10,00р.
75	Rhönbräu Klosterbier	7,75	125	25	ЛОЖЬ		
76	Lakalikczi	18	57	20	ЛОЖЬ		
77	Original Frankfurter grüne Soße	13	32	15	ЛОЖЬ		
	Общая стоимость товаров на складе:		74,050,85р.				
	Общая стоимость товаров к заказу:		3,633,45р.				

Рис. 2 Последние строки с итоговыми значениями

Решение

- Создайте новый файл Excel, на вкладке **Разработчик** в группе **Элементы управления** нажмите кнопку **Вставить** и среди элементов ActiveX щелкните по элементу управления **Кнопка** и нарисуйте кнопку на листе Excel. Для определенности будем считать, что созданная кнопка занимает ячейки с A1 по E1 первого листа.
- Щелкните по кнопке **Свойства** (проверьте, что нажата кнопка **Режим конструктора**) и в появившемся окне **Properties** настройте для свойства **Caption** значение **Получить данные**. При необходимости воспользуйтесь свойством **Font**, чтобы настроить подходящий шрифт для вашей кнопки.

- Щелкните правой кнопкой мыши по созданной вами кнопке и в контекстном меню выберите **Исходный текст**. Откроется редактор Visual Basic с курсором ввода на месте события Click для вашей кнопки.
- В окне редактора кода в меню **Tools** выберите **References** и установите флажок напротив строки Microsoft ActiveX Data Objects 2.1 Library.

Код для события Click вашей кнопки **Получить данные** может быть таким:

```
Private Sub CommandButton1_Click()

'Вначале — чистим всю книгу от старых данных
Cells.Select
Selection.Clear

' Создаем и настраиваем объект Connection
Dim cn As New ADODB.Connection
cn.ConnectionString = "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=C:\Nwind.mdb"
cn.Open

'Создаем и настраиваем объект Recordset
Dim rs As New ADODB.Recordset
rs.Open "SELECT[ProductID],[ProductName],[UnitPrice],[UnitsInStock],[ReorderLevel]," & _
"[Discontinued] FROM Products", cn

'На основе Recordset создаем объект QueryTable и вставляем его, начиная с 4-й строки
Dim QT1 As QueryTable
Set QT1 = QueryTables.Add(rs, Range("A4"))
QT1.Refresh

'Определяем количество записей в QueryTable
Dim nRowCount As Integer
Dim oRange As Range
Set oRange = QT1.ResultRange
nRowCount = oRange.Rows.Count

'Формируем столбец "Заказать товара, штук"
Range("G4").Value = " Заказать товара, штук "
Range("G4").Font.Bold = True
Range("G4").Columns.AutoFit

'Формируем столбец "Стоимость заказа"
```

```

Range("H4").Value = " Стоимость заказа"
Range("H4").Font.Bold = True
Range("H4").Columns.AutoFit
'Создаем диапазон, который включит в себя столбец G
' "вдоль" QueryTable
Set oRange = Range("G5", "G" & nRowCount + 3)
'Готовим переменные, которые нам потребуются в цикле
Dim oCell As Range
Dim sRowNumber As String
Dim cMoney As Currency
Dim cltogMoney As Currency
Dim cltogSklad As Currency
'Проходим циклом по всем ячейкам созданного диапазона
For Each oCell In oRange.Cells
'Получаем абсолютный номер строки в виде строковой переменной
sRowNumber = Replace(oCell.Address(True), "$G$", "")
'Проверяем определенные нами условия
If Range("E" & sRowNumber).Value > Range("D" & sRowNumber) And _
Range("F" & sRowNumber).Value = False Then
'Получаем значение для столбца G (заказ в штуках)
oCell.Value = (CInt(Range("E" & sRowNumber).Value) - CInt(Range("D" & sRowNumber).Value))
'Получаем значение для столбца H (стоимость заказа)
cMoney = (CInt(Range("E" & sRowNumber).Value) - CInt(Range("D" & sRowNumber).Value)) *
CCur(Range("C" & sRowNumber).Value)
'Записываем его в столбец H
Range("H" & sRowNumber).Value = cMoney
'Сразу плюсуем к итогу в рублях
cltogMoney = cltogMoney + cMoney
End If
'И в том же цикле сразу суммируем стоимость товаров на складе
cltogSklad = cltogSklad + (Range("C" & sRowNumber).Value * Range("D" & sRowNumber).Value)
Next

```

'Формируем две строки с итогами

```
Range("B" & nRowCount + 6).Value = " Общая стоимость товаров на складе:"
```

```
Range("B" & nRowCount + 6).Font.Bold = True
```

```
Range("B" & nRowCount + 7).Value = " Общая стоимость товаров к заказу:"
```

```
Range("B" & nRowCount + 7).Font.Bold = True
```

```
Range("D" & nRowCount + 6).Value = cltogSklad
```

```
Range("D" & nRowCount + 6).Font.Bold = True
```

```
Range("D" & nRowCount + 7).Value = cltogMoney
```

```
Range("D" & nRowCount + 7).Font.Bold = True
```

'Для красоты выделяем итоговое значение

```
Range("D" & nRowCount + 7).Select
```

'и производим скроллинг

```
Range("D" & nRowCount + 7).Show
```